
Answer set programming (ASP) (Brewka, Eiter, and
Truszczyński 2011) is a paradigm of declarative prob-
lem solving with roots in knowledge representation,

logic programming, constraint satisfaction and optimization.
Formally, ASP is based on the stable model semantics for log-
ic programs (Gelfond and Lifschitz 1991), detailed by Lif-
schitz (2016) in this issue. As illustrated by Janhunen and
Niemelä (2016), also in this issue, logic programs can be used
to compactly represent search and optimization problems
within the first two levels of the polynomial time hierarchy
(Faber, Pfeifer, and Leone 2011; Ferraris 2011).1 On the one
hand, the attractiveness of ASP is due to an expressive mod-
eling language, where concepts like (first-order) variables,
negation by default, and recursion enable uniform problem
representations in terms of facts specifying an instance along
with a general problem encoding (Schlipf 1995). On the oth-
er hand, powerful ASP systems, described by Kaufmann et al.
(2016) in this issue, are available off the shelf and automate
the grounding of an encoding relative to a problem instance
as well as the search for answer sets corresponding to prob-
lem solutions.

Modeling and 
Language Extensions

� Answer set programming (ASP) has
emerged as an approach to declarative
problem solving based on the stable
model semantics for logic programs.
The basic idea is to represent a compu-
tational problem by a logic program,
formulating constraints in terms of
rules, such that its answer sets corre-
spond to problem solutions. To this end,
ASP combines an expressive language
for high-level modeling with powerful
low-level reasoning capacities, provided
by off-the-shelf tools. Compact problem
representations take advantage of gen-
uine modeling features of ASP, includ-
ing (first-order) variables, negation by
default, and recursion. In this article,
we demonstrate the ASP methodology
on two example scenarios, illustrating
basic as well as advanced modeling and
solving concepts. We also discuss mech-
anisms to represent and implement
extended kinds of preferences and opti-
mization. An overview of further avail-
able extensions concludes the article.



In this article, we detail the ASP modeling method-
ology on two example scenarios. To begin with, we
elaborate on the use of traditional one-shot solving,
where a problem is tackled by means of singular
grounding and search processes. We particularly
focus on a conceptual generate-and-test pattern (Eit-
er, Ianni, and Krennwallner 2009; Leone et al. 2006;
Lifschitz 2002) as a best practice method to conceive
legible yet efficient problem encodings. Further infor-
mation regarding, among others, tool support for
logic program development, elaboration of tolerant
ways to represent extensive application domains, and
alternative modeling languages is provided by Lier-
ler, Maratea, and Ricca (2016), Erdem, Gelfond, and

Leone (2016), as well as Bruynooghe, Denecker, and
Truszczyński (2016) in this issue.

In our second example scenario, we take advantage
of multishot solving, a powerful extension of tradi-
tional ASP methods in which grounding and search
are interleaved to process a series of evolving subtasks
in an iterative manner. Rather than processing each
subtask from scratch, multishot solving gradually
expands the representation of a problem, where
grounding instantiates novel problem parts and
search can reuse conflict information. Such incre-
mental reasoning fits the needs in dynamic domains
like, for example, logistics, policies, or robotics. In
particular, we address a planning problem, where the

Figure 1. Places Connected by Links Associated with Costs.

Figure 2. The Shortest Round Trip for the Places in Figure 1.

(minimal) number of actions required to achieve a
goal is usually not known a priori, while theoretical
limits are prohibitively high as regards grounding.

The presentation of the two main approaches to
modeling and solving is complemented by a survey
of mechanisms to represent and implement extended
kinds of preferences and optimization. An overview
of further extensions conceived for demanding appli-
cation problems concludes the article.

Modeling the Traveling 
Salesperson Problem

For illustrating the principal modeling concepts in
ASP, let us consider the well-known traveling sales-
person problem (TSP). A TSP instance consists of a
number of places, each of which must be included
within a round trip visiting every place exactly once,
as well as links between places, specifying potential
successors along with associated costs.

For example, figure 1 displays an instance with six
places — the German cities Berlin, Dresden, Ham-
burg, Leipzig, Potsdam, and Wolfsburg, each denoted
by its first letter. The cities are linked by train con-
nections, available in either one or both directions,
and their respective durations in hours constitute the
costs. For example, Berlin and Potsdam are mutually
linked, and it takes one hour to travel between the
two neighboring cities, while four hours are needed
from Potsdam to Dresden or vice versa. Moreover, a
train connects Potsdam to Hamburg within three
hours, but it does not operate the other way round.
Further train connections interlink the other cities,
and the question is how to arrange a shortest round
trip visiting all cities.

When we construct a round trip manually, we may
first fix some place to start the trip from, say Pots-
dam, and then proceed by opportunistically picking
links to yet unvisited cities. for example, we can find
a round trip leading from Potsdam to Berlin, Ham-
burg, Leipzig, Wolfsburg, Dresden, and then back to
Potsdam. The connections taken in this trip add up
to a total duration of 1 + 2 + 2 + 1 + 2 + 4 = 12 hours.
However, the true shortest round trip shown in figure
2 takes only 11 hours. To find such shortest round
trips, and also in case the train connections or cities
to visit change, we aim at a general method for arbi-
trary places and links between them.2 In what fol-
lows, we thus apply the ASP methodology to model
shortest round trips for any TSP instance provided as
input.

Problem Instance
The common practice in ASP is to represent a prob-
lem at hand uniformly, distinguishing between a par-
ticular instance and a general encoding (Marek and
Truszczyński 1999, Niemelä 1999, Schlipf 1995). That
is, we first need to fix a logical format for specifying
places and links with associated costs. For example,

the cities and connections displayed in figure 1 are
described in terms of the facts given in listing 1.
These facts are based on two predicates, place/1 and
link/3, where 1 and 3 denote the arities of respective
relations. The letters used as arguments of facts over
place/1 stand for corresponding cities, for example, p
refers to Potsdam, and writing such constants in low-
ercase follows logic programming conventions.
Moreover, facts over link/3 specify the available con-
nections, for example, link(p, b, 1), link(p, d, 4), and
link(p, h, 3) provide those from Potsdam to Berlin,
Dresden, and Hamburg along with their associated
durations, as displayed in figure 1. The durations are
given by integers, on which ASP systems support
arithmetic operations,3 while the names used for
cities and predicates have no particular meaning
beyond identifying places or relations, respectively.
Also note that the facts constitute a set, so that the
order of writing them is immaterial, which distin-
guishes ASP from logic programming languages hav-
ing a procedural flavor, such as Prolog.

Problem Encoding
The main modeling task consists of specifying the
intended outcomes, that is to say, shortest round
trips, in terms of the conditions they must fulfill. To
this end, let us first formulate such requirements in
natural language: 

(a) Every place is linked to exactly one successor in a
trip. 
(b) Starting from an arbitrary place, a trip visits all
places and then returns to its starting point. 
(c) The sum of costs associated with the links in a trip
ought to be minimal. 

Listing 1. Instance Specifying the Places in Figure 1 as Facts.

1 place(b). % Berlin
2 place(d). % Dresden
3 place(h). % Hamburg
4 place(l). % Leipzig
5 place(p). % Potsdam
6 place(w). % Wolfsburg
7 link(b,h,2). link(b,l,2). link(b,p,1).
8 link(d,b,2). link(d,l,2). link(d,p,4).
9 link(h,b,2). link(h,l,2). link(h,w,3).
10 link(l,d,2). link(l,w,1).
11 link(p,b,1). link(p,d,4). link(p,h,3).
12 link(w,d,2). link(w,h,3). link(w,l,1).



While these conditions are sufficient to characterize
shortest round trips, the requirement in (b) further
implies that every place has some predecessor. Given
that (a) limits the number of links in a round trip to
the number of places, the following condition must
hold as well: 

(d) Every place is linked to exactly one predecessor in
a trip.

In summary, trips meeting the requirements in (a)
and (b) are subject to the optimality criterion in (c),
and (d) expresses an implied property. The condi-
tions at hand provide a mental model for the ASP
encoding furnished in the following.

The encoding shown in listing 2 is based on a con-
ceptual generate-and-test pattern (Eiter, Ianni, and
Krennwallner 2009; Leone et al. 2006; Lifschitz
2002). Accordingly, it is structured into several parts,
distinguished by their concerns as well as typical
constructs among those presented by Lifschitz (2016)
in this issue. The purposes of the parts indicated by
comments in lines beginning with % are as follows. 

A DOMAIN part specifies auxiliary concepts that can
be derived from facts and are shared by all answer sets.

A GENERATE part includes nondeterministic con-
structs, usually choice or disjunctive rules, to provide
solution candidates. 

A DEFINE part characterizes relevant properties of
solution candidates, where the inherent features of fix-
point constructions and negation by default suppress
false positives and enable a compact representation. 

A TEST part usually consists of integrity constraints
that deny invalid candidates whose properties do not
match the requirements on solutions. 

An OPTIMIZE part makes use of optimization state-
ments or weak constraints to associate solutions with
costs subject to minimization. 

A DISPLAY part declares output predicates to which
the printing of answer sets ought to be restricted in
order to make reading off solutions more convenient.

In what follows, we elaborate on respective encod-
ing parts.

DOMAIN
The first part, denoted by DOMAIN, includes the rule
in line 2 to determine the lexicographically smallest
identifier among places in an instance as (arbitrary)
starting point for the construction of a round trip. To
this end, the identifiers given by facts over place/1 are
taken as values for the variable Y, and the smallest
value, selected through a #min aggregate, is used to
instantiate the variable X recurring in the head
start(X). Note that, as usual in logic programming,
variable names begin with uppercase letters, and
recurrences within the same scope, that is, a rule, are
substituted with common values. Relative to the facts
in listing 1, the rule in line 2 is thus instantiated to

start(b) :- b = #min{

b : place(b); d : place(d); h : place(h); 

p : place(p); 1 : place(1); w : place(w)}.

Since the predicate place/1 is entirely determined by
facts, the above rule can be simplified to a derived
fact start(b). In general, a DOMAIN part contains
deterministic rules specifying relevant auxiliary con-
cepts, so that they do not need to be provided per
instance in a redundant fashion. Rather, including
such rules in an encoding increases elaboration tol-
erance and exploits the capabilities of common
grounders, which evaluate deterministic parts.

GENERATE
The second part, indicated by GENERATE, gathers
nondeterministic constructs such that alternative
selections among the derivable atoms provide dis-
tinct solution candidates. In line 4, we use a choice
rule (Simons, Niemelä, and Soininen 2002) to express
that, for every place in an instance, exactly one link
from the place must be picked for a round trip. The
rule constitutes a schema that applies to each place
identifier taken as value for the variable X. For exam-
ple, considering Potsdam and the three connections
from there, it yields

{travel(p, b) : link(p, b, 1); 
travel(p, d) : link(p, d, 4); 
travel(p, h) : link(p, h, 3)} = 1 :- place(p).

Listing 2. Encoding of Round Trips with
Regard To Facts as in Listing 1.

Apart from a system-specific #show directive for output projection, the
encoding in listing 2 is written in the syntax of the ASP-Core-2 standard
language (www.mat.unical.it/aspcomp2013/ASPStandardization).

1 % DOMAIN
2  start(X) :- X = #min{Y : place(Y)}.
3  % GENERATE
4  {travel(X,Y) : link(X,Y,C)} = 1 :- place(X).
5  % DEFINE
6  visit(X)  :- start(X).
7  visit(Y)  :- visit(X), travel(X,Y).
8  % TEST
9  :- place(Y),  not visit(Y).
10  :- start(Y),  #count{X : travel(X,Y)} < 1.
11  :- place(Y),  #count{X : travel(X,Y)} > 1.
12  % OPTIMIZE
13  :~travel(X,Y), link(X,Y,C). [C,X]
14  % DISPLAY
15  #show travel/2.



Further simplifying this rule in view of facts over
place/1 and link/3 leads to

{travel(p, b); travel(p, d); travel(p, h)} = 1.

That is, any answer set must include exactly one of
the options travel(p, b), travel(p, d), and travel(p, h),
reflecting that either Berlin, Dresden, or Hamburg
has to succeed Potsdam in a round trip. As the same
schema applies to other cities as well, atoms over the
predicate travel/2 in an answer set represent a trip
meeting the requirement in (a). However, the rule in
line 4 leaves open which successor per place shall be
picked, and hence it is called choice rule.

DEFINE
While the predicate travel/2 provides sufficient infor-
mation to reconstruct a trip from an answer set, the
requirement in (b) that all places must be visited is
yet unaddressed. In order to test this condition, the
DEFINE part includes rules analyzing which places
are visited from the starting point fixed in the
DOMAIN part before. To begin with, the rule in line
6 derives the starting point as visited, for example,
visit(b) follows from start(b) relative to the facts in list-
ing 1. The rule in line 7 further collects places reach-
able through the connections indicated by travel /2.
For example, the following derivation chain is acti-
vated by atoms over travel/2 that represent the con-
nections shown in figure 2:

visit(b) :- start(b).

visit(p) :- visit(b), travel(b, p).

visit(h) :- visit(p), travel(p, h).

visit(l) :- visit(h), travel(h, l).

visit(w) :- visit(l), travel(l, w).

visit(d) :- visit(w), travel(w, d).

Given that the involved connections form a round
trip, all atoms over visit/1 follow through a sequence
of rules rooted in start(b). However, if Hamburg were
linked to Berlin instead of Leipzig, no such sequence
would yield visit(l), visit(w), and visit(d). Atoms lacking
a noncircular derivation are unfounded and exempt
from answer sets (Van Gelder, Ross, and Schlipf
1991). In turn, answer sets encompass fix-point con-
structions for expressing concepts like, for example,
induction and recursion. A DEFINE part makes use of
this to derive predicates indicating relevant proper-
ties of a solution candidate at hand. As in DOMAIN
parts, the contained rules are deterministic, yet their
evaluation relies on nondeterministically generated
solution candidates. In our case, visit/1 provides all
places reached by taking connections in the trip from
a fixed starting point.

TEST
The predicates characterizing solution candidates as
well as their relevant properties are inspected in the
TEST part in order to eliminate invalid candidates.
This is accomplished by means of integrity con-
straints, that is, rules of denial with an implicitly false
head, written by leaving the left side of :- blank.
Regarding the conditions for round trips, the GEN-

ERATE part already takes care of (a), while the
requirement in (b) remains to be checked. To this
end, the integrity constraint in line 9 expresses that
all places must be visited from the starting point giv-
en by start/1. for example, if Leipzig were not
reached, a contradiction would be indicated through
:- place(l), not visit(l). Note that not visit(l) makes use
of negation by default, which applies whenever vis-
it(l) is unfounded. Importantly, negation by default
does not offer any derivation (by contraposition). As
a consequence, the above integrity constraint is not
interchangeable with a rule like visit(l) :- place(l).

If given such a rule, we could simply conclude vis-
it(l), regardless of reachability. Unlike that, integrity
constraints do not modify solution candidates or
predicates providing their properties, but merely
deny unintended outcomes. The distinction
between constructs for deriving and evaluating
atoms is an important modeling concept, here used
to check that all places are indeed reached from a
fixed starting point.

For the requirement in (b), we still have to make
sure that a trip at hand returns to its starting point.
Since every place is linked to one successor only and
all but one final connection are needed to visit places
different from the starting point given by start/1, it is
sufficient to check that a (final) connection return-
ing to the starting point exists. This condition is
imposed by the integrity constraint in line 10, and
relative to the facts in listing 1 it is instantiated to

:- start(b), #count{d : travel(d, b); 
h : travel(h, b); p : travel(p, b)} < 1.

The #count aggregate provides the number of atoms
among travel(d, b), travel(h, b), and travel(p, b), repre-
senting connections returning to Berlin, included in
an answer set. If neither connection is taken, this
number is zero, in which case the success of the < 1
comparison indicates a contradiction. In turn, some
connection must lead back to Berlin, but it can only
be taken once all places are visited.

The checks through the integrity constraints in
lines 9 and 10 establish that answer sets represent
round trips meeting the requirement in (b). Since (a)
is handled in the GENERATE part, the rules up to line
10 are already sufficient to characterize round trips.
However, the implied property in (d) also states that
a place cannot be linked to several predecessors.
While this condition may seem apparent to humans,
it relies on a counting argument taking the number
of connections in a trip and the necessity that every
place must be linked to some predecessor into
account. Given that ASP solvers do not apply such
reasoning, it can be beneficial to formulate nontriv-
ial implied properties as redundant constraints. This
is the motivation to include the integrity constraint
in line 11, making explicit that a place cannot be
linked to several predecessors. For example, regard-
ing connections leading to Berlin, the schema yields

:- place(b), #count{d : travel(d, b); 

h : travel(h, b); p : travel(p, b)} > 1.

In view of the > 1 comparison relative to the #count
aggregate, a contradiction is indicated as soon as con-
nections from two cities among Dresden, Hamburg,
and Potsdam to Berlin are picked for a round trip.
Respective restrictions to a single predecessor apply
to cities other than Berlin as well.

OPTIMIZE
After specifying solution candidates and require-
ments, the OPTIMIZE part addresses the optimality
criterion in (c). To this end, the weak constraint in
line 13 associates every place with the cost of the link
to its successor in a round trip.4 Focusing on the
three connections from Berlin, we obtain

:˜ travel(b, h), link(b, h, 2). [2, b]
:˜ travel(b, l), link(b, l, 2). [2, b]
:˜ travel(b, p), link(b, p, 1). [1, b]

Weak constraints resemble integrity constraints, but
rather than eliminating solution candidates to which
the expressed conditions apply, the lists enclosed in
square brackets are gathered in a set. The sum of inte-
gers included as their first elements constitutes the
total cost associated with an answer set and is subject
to minimization. Regarding connections from Berlin,

the fraction of the total cost is either 1 for Potsdam or
2 in case of Hamburg and Leipzig. Given that Ham-
burg and Leipzig cannot both succeed Berlin in a
round trip, there is no urge to keep their respective
lists distinct, for example, by adding the identifier h
or l as an element. By reusing the same list instead, we
actually reduce the number of factors taken into
account in the total cost calculation, which can in
turn benefit the performance of ASP solvers.5

DISPLAY
The final part, denoted by DISPLAY, includes the
#show directive in line 15, declaring travel/2 as out-
put predicate. This does not affect the meaning of the
encoding, but instructs systems like clingo (Gebser et
al. 2014) to restrict the printing of answer sets to
atoms over travel/2. Indeed, facts as well as places giv-
en by start/1 and visit/1 are predetermined by an
instance, and only the connections provided by trav-
el/2 characterize a particular round trip.

Solution Computation
Assuming that the facts in listing 1 and the encoding
in listing 2 are stored in text files called tsp-ins.lp and
tsp-enc.lp, the output of a clingo run is given in list-
ing 3. We see that clingo finds three round trips of
decreasing cost, listed in lines beginning with Opti-
mization: below the atoms over travel/2 in a corre-
sponding answer set. While the first round trip is
arbitrary and merely depends on heuristic aspects of
the search in clingo, the second must be of smaller
cost, and likewise the third. The latter cannot be
improved any further, indicated by OPTIMUM FOUND
in the last line, as it represents the shortest round trip
shown in figure 2. For the instance at hand, this is the
only round trip of cost 11, and some arbitrary witness
among all optimal answer sets is determined in gen-
eral.6 Nondeterminisms, such as the (optimal) answer
set found, are thus left up to the search in an ASP
solver, while an encoding merely specifies require-
ments on intended outcomes. This distinguishes ASP
from traditional logic programming languages like
Prolog, in which programs have a procedural seman-
tics based on the order of writing rules.

Summary
While the well-known TSP is conceptually simple, it
gives room for exploring diverse modeling concepts
and designs. Let us recap the main principles of the
above ASP method.

A uniform problem representation separates facts
describing an instance from a general problem
encoding. The latter consists of schemata, expressed
in terms of variables, that specify solutions for any
problem instance. Such high-level modeling is crucial
for elaboration tolerance, meaning that changes in a
problem specification can be addressed by modest
modifications of the representation. For example,
when round trips shall be approximated for instances

Listing 3. clingo Run on Facts and Encoding in Listing 1-2.

1 $ clingo tsp-ins.lp tsp-enc.lp

3 Answer: 1
4  travel(b,l) travel(l,w) travel(w,d)
5  travel(d,p) travel(p,h) travel(h,b)
6 Optimization: 14

8 Answer: 2
9  travel(b,p) travel(p,h) travel(h,w)
10  travel(w,l) travel(l,d) travel(d,b)
11 Optimization: 12

13 Answer: 3
14  travel(b,p) travel(p,h) travel(h,l)
15  travel(l,w) travel(w,d) travel(d,b)
16 Optimization: 11

18 OPTIMUM FOUND



that have no solution otherwise, the integrity con-
straint requiring all places to be visited can easily be
turned into a weak constraint for admitting excep-
tions.

An ASP encoding is usually structured into parts
addressing different concerns in a generate and test
conception. The key parts, nicknamed GENERATE,
DEFINE, TEST, and OPTIMIZE, provide solution can-
didates, analyze their relevant properties, eliminate
invalid candidates, and evaluate solution quality.

The typical constructs used within the GENERATE,
DEFINE, TEST, and OPTIMIZE parts are nondeter-
ministic (choice) rules, deterministic rules, integrity
constraints, or weak constraints, respectively. Deter-
ministic rules make use of the expressivity of answer
sets encompassing fix points, induction, and recur-
sion. While TEST parts should typically stay compact
regarding sufficient conditions, redundant con-
straints expressing nontrivial implied properties can
benefit the search in an ASP solver. Weak constraints
in an OPTIMIZE part can be made more effective by
reducing the number of factors taken into account to
evaluate solution quality.

An ASP encoding merely specifies requirements,
but not how answer sets representing (optimal) solu-
tions shall be computed. While admissible outcomes
are fixed by the semantics, the way to find them is
left up to ASP solvers.

Modeling the Blocks World 
Planning Problem

Beyond traditional one-shot solving, where a prob-
lem instance is fed to an isolated search process, mul-
tishot solving addresses series of evolving subtasks in
an iterative manner. This is of interest in dynamic
domains, such as logistics, policies, or robotics, deal-
ing with recurrent tasks in a changing environment.
To illustrate respective scenarios, we consider blocks
world planning (Slaney and Thiébaux 2001), where
blocks must be restacked on a table to bring them
from their initial positions into a goal configuration.

Figure 3 displays an example scenario with nine
blocks. In the initial situation, shown on the left, the
blocks are arranged in three stacks, and the two
stacks on the right constitute the goal situation. To
change the configuration, a free block at the top of
some stack can be moved on top of another stack or
to the table. That is, a block offers room for at most
one other block on top of it, while any number of
blocks can be put on the table. A naive strategy to
establish the goal situation thus consists of succes-
sively moving all blocks to the table, and then build
up required stacks from the bottom. For the dis-
played scenario, this results in six moves to the table
plus seven moves to construct the stacks on the right.
However, the interest is to perform as few moves as

Figure 3. Initial and Goal Situation for Blocks World Planning.

needed, and in the following we show how shortest
plans can be found using multishot solving.

Problem Instance
Similar to one-shot solving, applied to the TSP before,
a problem instance is described in terms of facts.
Those representing the situations displayed in figure
3 are given in listing 4, where the predicates init/2
and goal/2 specify the respective positions of blocks.
In addition, table(0) declares 0 as identifier for the
table, which is at the bottom of stacks in both the ini-
tial and the goal configuration.

Problem Encoding
To exploit the multishot solving capacities provided
by the clingo system (Gebser et al. 2014), the encod-
ing given in listing 5 is composed of three subpro-
grams. Their names and parameters are introduced
by #program directives, and a subprogram includes
the rules up to the next such directive (if any). In the
context of planning, the subprograms are dedicated
to the following concerns:

(1) A base subprogram is processed once for pro-
viding auxiliary concepts along with setting up an
initial configuration. (2) A check(t) subprogram is
parametrized by a constant t, serving as a placehold-
er for successive integers starting from 0. For each
time point taken as a value to replace t with, integri-
ty constraints impose goal conditions. They include
a dedicated atom query(t), provided by clingo for the
current last time point, while obsolete conditions are
deactivated to reflect an increased plan length. (3) A
step(t) subprogram is likewise parametrized, yet t is
replaced with successive integers starting from 1. This
subprogram specifies transitions in terms of rules for
picking actions, deriving atoms that represent a suc-
cessor configuration, and asserting the validity of a
transition. In contrast to check(t), such rules are not
withdrawn but joined with others obtained at later
time points.

The subprograms are further structured into con-
ceptual DOMAIN, GENERATE, DEFINE, and TEST
parts. Moreover, the DISPLAY part declares move/3 as
output predicate (for all subprograms) through the
#show directive in line 29, while the solving process
of clingo focuses on shortest plans without requiring
any OPTIMIZE part.

base
The first subprogram, called base, contributes a
DOMAIN part consisting of the rules from lines 3 to
5. The idea of the predicate do/2 is to provide moves
that could be relevant to a shortest plan. In particu-
lar, the rule in line 3 expresses that moving a block to
the table can be useful for accessing the stack under-
neath, but only if such a stack exists and the block is
not already on the table in the initial situation. Giv-
en the stacks on the left in figure 3, we thus derive
that the blocks numbered 2, 3, 5, 6, 8, and 9 may be
moved to the table. In addition, the rule in line 4

Listing 5. Blocks World Encoding with Regard to Facts as in Listing 4.

1 #program base.
2 % DOMAIN
3 do(X,Z) :- init(X,Y), not table(Y), table(Z).
4 do(X,Y) :- goal(X,Y), not table(Y).
5 on(X,Y,0) :- init(X,Y).

7 #program check(t).
8 % TEST
9 :- query(t), goal(X,Y),  not on(X,Y,t).
11 #program step(t).
12 % GENERATE
13 {move(X,Y,t) : do(X,Y)} = 1.
14 % DEFINE
15 move(X,t)  :- move(X,Y,t).
16 on(X,Y,t)  :- move(X,Y,t).
17 on(X,Y,t)  :- on(X,Y,t–1), not move(X,t).
18 lock(Y,t)  :- on(X,Y,t–1), not table(Y).
19 firm(X,t)  :- on(X,Y,t), table(Y).
20 firm(X,t)  :- on(X,Y,t), firm(Y,t).
21 % TEST
22 :- lock(X,t), move(X,t).
23 :- lock(Y,t), move(X,Y,t).
24 :- init(Y,Z), #count{X : on(X,Y,t)} > 1.
25 :- init(X,Z), #count{Y : on(X,Y,t)} > 1.
26 :- init(X,Z), not firm(X,t).

28 % DISPLAY
29 #show move/3.

Listing 4. Instance Specifying Situations in Figure 3 as Facts.

1 init(3,2). init(6,5). init(9,8).
2 init(2,1). init(5,4). init(8,7).
3 init(1,0). init(4,0). init(7,0). table(0).
4 goal(8,6).
5 goal(6,4). goal(5,7).
6 goal(4,2). goal(7,3).
7 goal(2,1). goal(3,9).
8 goal(1,0). goal(9,0).



indicates moves to goal positions different from the
table. Regarding the goal configuration on the right
in figure 3, we obtain corresponding moves for all
blocks but those numbered 1 and 9. As a result,
derived facts over do/2 yield at most two relevant
moves per block, while other moves may be legal but
cannot belong to shortest plans.7 The remaining rule
in line 5 maps initial positions to derived facts over
on/3, where the integer 0 denotes a time point asso-
ciated with the initial configuration.

check(t)
The subprogram check(t) is parametrized by a con-
stant t that is handled by clingo and replaced with
successive integers starting from 0. It contributes a
TEST part, including the integrity constraint in line 9,
to deny plans such that some goal position is not yet
established at the last time point referred to by t. This
is accomplished by means of a dedicated atom
query(t), provided by clingo for the current last time
point and deactivated when proceeding to the next
integer to replace twith. For example, the initial posi-
tion of block 3 on the left in figure 3 does not match
its goal position on the right, and a contradiction is
indicated through

:- query(0), goal(3, 9), not on(3, 9, 0).

However, query(0) holds only as long as 0 is the last
time point, while query(1) is used for 1 instead, and
so on.

step(t)
The third subprogram, denoted by step(t), specifies
transitions to time points referred to by its parameter
t, serving as a placeholder for successive integers
starting from 1. To begin with, the choice rule in line
13 constitutes the GENERATE part for picking one
among the moves taken as relevant in the base sub-
program. Note that the current time point is used as
third argument in atoms over move/3, while do/2
remains fixed, regardless of the time point.

The deterministic rules in the DEFINE part from
line 15 to 20 derive further atoms characterizing a
transition at hand. A block changing its position is
extracted through projection to move/2. Atoms over
on/3, representing a successor configuration, are
derived from a move as well as inertia applying to all
blocks but the one that is moved. Again harnessing
projection, the predicate lock/2 indicates blocks that
were not on top of a stack and can thus not partici-
pate in legal moves. Finally, the predicate firm/2 pro-
vides blocks rooted on the table in a successor con-
figuration, where noncircular derivations similar to
those for places reachable in the TSP have the table as
their starting point.

The TEST part, including the integrity constraints
from line 22 to 26, then eliminates inexecutable
plans. Moves involving inaccessible blocks are ruled
out in line 22 and 23, which is actually sufficient to
check that a plan can be executed.

Notably, the first of these integrity constraints
reuses the projection to move/2, as only the moved

block is of interest here. In addition, line 24 to 26
impose redundant state constraints, making explicit
that, in any configuration, no block is under or on
several objects and all blocks are rooted on the table.8

For example, this expresses that block 3 cannot be at
its goal position in between the blocks numbered 7
and 9 as long as the third stack displayed on the left
in figure 3 is intact, no matter the performed moves.

Solution Computation
The output of clingo run on the facts in listing 4 and
the encoding in listing 5, stored in text files blocks-
ins.lp and blocks-enc.lp, is given in listing 6. The 10
lines saying Solving... indicate that 10 time points,
namely successive integers from 0 to 9, have been
used for the parameter of the check(t) subprogram.
Apart from 0, they are also applied to the step(t) sub-
program describing transitions, while base is
processed just once at the beginning. Failed attempts
to find an answer set for time points from 0 to 8
mean that there is no plan consisting of a respective
number of moves. In turn, the plan found for time
point 9 is shortest. The contained atoms over move/3
mainly convey that moving the blocks numbered 6,
8, and 9 to the table allows for building up the goal
stacks. Alternative shortest plans, which can be
obtained by enumerating answer sets, include a
move of block 5, rather than block 8, to the table.

Listing 6. clingo Run on Facts and Encoding in Listings 4 and 5.

1 $ clingo blocks-ins.lp blocks-enc.lp

3 Solving...
4  Solving...
5  Solving...
6  Solving...
7  Solving...
8  Solving...
9  Solving...
10  Solving...
11  Solving...
12  Solving...

14  Answer: 1
15     move(9,0,1) move(6,0,2) move(3,9,3)
16     move(8,0,4) move(7,3,5) move(5,7,6)
17     move(4,2,7) move(6,4,8) move(8,6,9)



Summary
The blocks world is a dynamic domain, in which
actions change the state of the environment over
time. Shortest plans to progress from an initial to a
goal situation can be found using multishot solving
according to some basic principles:

An instance is provided by facts specifying the objects
of interest along with initial and goal conditions.

A general problem encoding furnishes three subpro-
grams, called base, check(t), and step(t). The latter are
parametrized by a constant, here denoted t, serving as
a placeholder for successive integers starting from 0 or
1, respectively.

The base subprogram is processed once at the begin-
ning. It typically contributes a DOMAIN part setting
up auxiliary concepts as well as atoms representing an
initial configuration.

Occurrences of parameter t in the check(t) subprogram
are successively replaced with integers from 0. The
common purpose is to impose goal conditions by
means of integrity constraints in a TEST part. By using
a dedicated atom query(t) in integrity constraints,
obsolete conditions are deactivated when proceeding
to the next integer.

The step(t) subprogram is processed analogously to
check(t), yet starting from integer 1 instead of 0. This
predestinates step(t) to specify the transition to a suc-
cessor configuration associated with t. The constructs
typical for the GENERATE, DEFINE, and TEST parts are
used to provide candidates, derive atoms characteriz-
ing them, and eliminate invalid transitions. Invariant
properties can be expressed by incorporating redun-
dant state constraints.

While facts as well as the base subprogram are
processed only at the beginning, multishot solving by
clingo iteratively adds rules obtained by replacing the
parameters of the check(t) and step(t). subprograms
with successive integers. This corresponds to gradual-
ly increasing the plan length until an answer set rep-
resenting a shortest plan is found. The required length
is often not known a priori, and multishot solving
allows for discovering it.

Preferences and Optimization
The identification of preferred, or optimal, solutions
is often indispensable in real-world applications, as
illustrated on the TSP and blocks world scenarios
above. In many cases, this also involves the combi-
nation of various qualitative and quantitative prefer-
ences. In fact, optimization statements representing
objective functions based on summation or counting
are integral concepts of ASP systems since their
beginnings (manifested by #minimize and #maxi-
mize statements [Simons, Niemelä, and Soininen
2002] or weak constraints [Leone et al. 2006]). The
built-in repertoire of current ASP systems also covers
set-inclusion-based optimization (Gebser et al. 2015).

Other approaches to optimizing relative to specific
and often more complex types of preference are fur-

nished by dedicated external systems. Such
approaches can be categorized into two classes (com-
pare Delgrande et al. [2004]). On the one hand, we
find prescriptive approaches to preference that take
an order on rules and then enforce this order during
the construction of optimal answer sets (Brewka and
Eiter 1999). Such prescriptive approaches do not lead
to an increase in computational complexity, which
makes them amenable to implementation by compi-
lation (Delgrande, Schaub, and Tompits 2003) or
metainterpretation (Eiter et al. 2003). On the other
hand, we have descriptive approaches that impose
preferences among the answer sets of a program
(Brewka, Niemelä, and Truszczyński 2003; Sakama
and Inoue 2000; Son and Pontelli 2006). Unlike the
former, these approaches typically lead to an elevat-
ed level of complexity, which makes their efficient
implementation more challenging. The asprin sys-
tem (Brewka et al. 2015) offers a general and flexible
framework for computing optimal answer sets rela-
tive to preferences among them.9 In particular, its
library comprises all afore-cited descriptive approach-
es and further allows for freely combining prefer-
ences of qualitative and quantitative nature.

Further Extensions
The previous sections presented some popular mod-
eling features and extensions, for example, relative to
propositional satisfiability (SAT), going along with
the ASP methodology. These include uniform prob-
lem representations using (first-order) variables with-
in encodings, aggregates expressing collective condi-
tions on sets, optimization, and multishot solving
capacities. While such concepts already provide rich
facilities for modeling and solving complex compu-
tational problems, we conclude with a (nonexhaus-
tive) overview of further extensions.

Similar to disjunctive rules, nonmonotone recur-
sive aggregates (Faber, Pfeifer, and Leone 2011; Fer-
raris 2011) allow for expressing problems at the sec-
ond level of the polynomial time hierarchy.
Finite-domain constraints specifying quantitative
conditions can be addressed through dedicated back-
ends (Aziz, Chu, and Stuckey 2013; Balduccini 2011;
Mellarkod, Gelfond, and Zhang 2008; Ostrowski and
Schaub 2012) or compilation (Banbara et al. 2015;
Drescher and Walsh 2010). Moreover, translation
approaches allow for handling real numbers
(Bartholomew and Lee 2013; Liu, Janhunen, and
Niemelä 2012). Extended functionalities like multi-
shot solving are realized by combining ASP systems
with scripting languages (Gebser et al. 2014). Further
details regarding the integration of ASP systems with
imperative languages or external information sources
are provided by Lierler, Maratea, and Ricca (2016)
and Erdem, Gelfond, and Leone (2016) in this issue.
As also discussed in the latter article, high-level prob-
lem representations, for example, specified in terms

of action languages, can in turn be mapped to ASP
through corresponding front ends.

Notes
1. See, for example, Papadimitriou (1994) for an introduc-
tion to computational complexity.

2.  Computing a shortest round trip is FPNP-complete
(Papadimitriou 1994); that is, it can be accomplished by
means of a polynomial number of queries to an NP-oracle.

3.  Extensions to real numbers are presented by
Bartholomew and Lee (2013) and Liu, Janhunen, and
Niemelä (2012).

4. The weak constraint corresponds to the optimization
statement #minimize {C, X : travel(X, Y), link(X, Y, C)}.

5. An even more elaborate penalization scheme based on
relative cost differences is presented by Gebser et al. (2012,
section 8.3).

6. Optimal answer sets can be enumerated using dedicated
reasoning modes of clingo (Gebser et al. 2015).

7. More elaborate conditions to further restrict potential
moves are provided by Slaney and Thiébaux (2001), and
respective ASP encodings are presented by Gebser et al.
(2012, section 8.2). While such domain knowledge as well
as the encoding in listing 5 are specific to Blocks World
Planning, domain-independent approaches to model
actions and change are discussed by Erdem, Gelfond, and
Leone (2016) in this issue.

8. Similar constraints are also included in encodings pre-
sented by Erdem and Lifschitz (2003), Gebser et al. (2012),
and Lifschitz (2002) and further pave the way to partially
ordered plans with parallel actions.

9. The only requirement is that evaluating a preference
must be encodable in ASP (and thus have a complexity not
beyond the second level of the polynomial time hierarchy)

.

Ianni, and M. Truszczyński, 368–383. Berlin: Springer.
dx.doi.org/10.1007/978-3-319-23264-5_31

Gebser, M.; Kaminski, R.; Kaufmann, B.; and Schaub, T.
2012. Answer Set Solving in Practice. Synthesis Lectures on
Artificial Intelligence and Machine Learning. San Rafael,
CA: Morgan and Claypool Publishers.

Gebser, M.; Kaminski, R.; Kaufmann, B.; and Schaub, T.
2014. Clingo = ASP + Control: Preliminary Report. In Tech-
nical Communications of the Thirtieth International Conference
on Logic Programming (ICLP-14), Theory and Practice of Log-
ic Programming, Online Supplement .arXiv Volume:
1405.3694v1. Ithaca, NY: Cornell University Library.

Gelfond, M., and Lifschitz, V. 1991. Classical Negation in
Logic Programs and Disjunctive Databases. New Generation
Computing 9(3/4): 365–385. dx.doi.org/10.1007/BF03037169

Janhunen, T., and Niemelä, I. 2016. The Answer Set Pro-
gramming Paradigm. AI Magazine 37(3).

Kaufmann, B.; Leone, N.; Perri, S.; and Schaub, T. 2016.
Grounding and Solving in Answer Set Programming. AI
Magazine 37(3).

Leone, N.; Pfeifer, G.; Faber, W.; Eiter, T.; Gottlob, G.; Perri,
S.; and Scarcello, F. 2006. The DLV System for Knowledge
Representation and Reasoning. ACM Transactions on Com-
putational Logic 7(3): 499–562. dx.doi.org/10.1145/1149114.
1149117

Lierler, Y.; Maratea, M.; and Ricca, F. 2016. Systems, Engi-
neering Environments, and Competitions. AI Magazine
37(3).

Lifschitz, V. 2002. Answer Set Programming and Plan Gen-
eration. Artificial Intelligence 138(1–2): 39–54.
dx.doi.org/10.1016/S0004-3702(02)00186-8

Lifschitz, V. 2016. Answer Sets and the Language of Answer
Set Programming. AI Magazine 37(3).

Liu, G.; Janhunen, T.; and Niemelä, I. 2012. Answer Set Pro-
gramming via Mixed Integer Programming. In Proceedings of
the Thirteenth International Conference on Principles of Knowl-
edge Representation and Reasoning (KR-12), ed. G. Brewka, T.
Eiter, and S. McIlraith, 32–42. Palo Alto, CA: AAAI Press.

